

PharmaSUG China 2015-47

Handling multiple y axes using SAS® Graphs

Mina Chen, Roche Product Development in Asia Pacific, Shanghai, China

ABSTRACT

Graphics are a powerful way to display clinical trial data. Graphs are widely generated in the pharmaceutical industry to review the results of clinical trials. For example, mean plot is often used to show laboratory results change over time. Sometimes there's a need to display the changes of two laboratory parameters over time within one mean plot. A second y-axis is useful for plotting data with different scales on the same plot.

This paper will show, by example, how such graphs can easily be created using the SAS/GRAPH®.

INTRODUCTION

Clear and informative graphics are a popular way to explore, analyze and display data in clinical trials. Graphical representation of drug safety or efficacy is typically easier to understand compared to a statistical analysis table. In many cases it is useful if not necessary, to add secondary y-axes to display data with different scales on the same plot, for example graphing two correlated lab parameters. With the release of SAS® 9.2 there are a variety of new options to create sophisticated analytical graphs. Such graph can be created using PROC TEMPLATE and SAS/GRAPH® SG procedures.

The goal of this paper is to show how to create graphs with multiple y-axes commonly used in the clinical trials domain using PROC TEMPLATE and SAS/GRAPH® SG procedures. In this paper, we will take sample graphs commonly used for analysis of clinical trials data,; mean plot, graphs for viewing laboratory tests, etc., and show the code needed to build each graph in detail.

Three types of graphs with multiple Y axes that are commonly used in clinical studies will be introduced. SAS 9.4 is used in this paper.

TYPE 1 –MEAN LINE PLOT WITH TWO Y AXES

For trials with a quantitative outcome measure it is common to have repeated measures at fixed follow-up times, and usually also at baseline. It is then usual to plot the means of test results over time.

Figure 1 plots the mean of laboratory measurements ALT and RNA changes at each time point. This graph presents RNA and ALT on the right and left Y axes respectively for each time point estimate, to illustrate the relationship of two measurements.

Such graph can be produced with PROC SGPLOT using the Y2AXIS statements. In this instance different data columns are mapped to the Y and Y2 axes. In the following example layout block, even though the Y and Y2 columns are different, the primary and secondary Y axes represent the same data range in different units.





Figure 1

Generally, the SG procedures contain 4 types of axis statements that enable you to change the type and appearance of your graph:



Below is sample code that illustrates the syntax. The Y2AXIS option specifies that the plot of mean RNA will be on a second Y axis. The values of the second Y axis are displayed on the right side of the plot.

```
proc sgplot data=ana.testa;
  scatter x=VISIT y=m_alt / markerattrs=(symbol=trianglefilled color=red) ;
  scatter x=VISIT y=m_rna / y2axis
                                markerattrs=(symbol=circlefilled color=blue);
  series x=VISIT y=m_alt / curvelabel='ALT'
                                lineattrs=(color=red pattern=dash) ;
  series x=VISIT y=m_rna / curvelabel='RNA'
                                Y2axis
                                lineattrs=(color=blue ) ;
  xaxis display=(nolabel);
  yaxis min=0 label='ALT ( U/L )' values=(0 to 100 by 10);
  y2axis min=0 label='RNA (log10 IU/ML )' values=(0 to 10 by 1);
run;
```

TYPE 2 – LINE PLOT AND BAR CHART IN A GINGLE PLOT WITH TWO Y-AXES

In analyzing clinical trial data, it is often valuable to visualize the changes in the efficacy parameter with respect to changes in vital signs measurements. In Figure 2, a bar chart to illustrate the distribution of weight or height for age percentile over time and a line plot to represent the change in biomarker of interest is plotted on the same graph. This graph illustrates the relationship between mean values of the weight/height for age percentile and an interested biomarker. The main feature of the plot includes two different Y-scales: the scale on the left axis represents the mean change in value of the biomarker of interest as a line plot and the scale on the right axis represents the weight/height for age percentiles over time. The horizontal axis is common to the two plots.




Figure 2

The code below uses the TEMPLATE and SGRENDER procedures to generate bar chart and line plot within one graph. The TEMPLATE procedure defines the statgraph template. When submitted, the syntax is compiled into a statgraph template called htwtpercentile. The YAXIS=Y2 option is specified in the two BARCAHRT statements, producing the right vertical axis. In this example, the Y2AXISOPTS statement is used to define the attributes of the second Y axis, e.g. value range, label etc. To produce the graph, the SGRENDER procedure must be used to associate the template with the analysis data set.

```

proc template;
  define statgraph htwtpercentile;
    begingraph;
      entrytitle ' ';
      layout overlay / cycleattrs=true
        xaxisopts=( label="Week")
        y2axisopts=( offsetmin=0 label="Weight/Height for Age Percentile:"
          linearopts=(tickvaluesequence=(start=0 end=150
            increment=15) viewmin=0 viewmax=150))
        yaxisopts=( offsetmin=0 label='Biomarker parameter'
          linearopts=(tickvaluesequence=(start=0 end=20
            increment=2) viewmin=0 viewmax=20));
      barchart x=visitn y=wt / stat=mean name="Wtpct"
        legendlabel="Wtpct"
        discreteoffset=-0.3 barwidth=0.3 yaxis=y2; ;
      barchart x=visitn y=ht / stat=mean name="Htpct"
        legendlabel="Htpct"
        discreteoffset=0 barwidth=0.3 yaxis=y2;
      discretelegend "Wtpct" "Htpct" / title=""
        location=inside halign=right valign=top;
      seriesplot x=visitn y=lbval / lineattrs=(pattern=solid thickness=2) ;
      scatterplot x=visitn y=lbval / ;
    endlayout;
  endgraph;
end;
run;

proc sgrender data=lbtest template=htwtpercentile;
run;

```

TYPE 3 – PANEL PLOTS WITH SECONDARY Y AXES

Panel plots are designed to produce graphs based on classification levels. Data can easily be compared across your classification variable thus making your graph more effective. Figure 3 shows the weight/ height for age

percentile changes over time among different treatment groups. The weight and height for age percentile are put in left and right Y axes separately. Notice that the data ranges for each Y axes are different, you can adjust the range based on study needs.




Figure 3

The following SAS code is used to create figure 3. In the TEMPLATE procedure, we defined a template called WHTHPANEL. The LAYOUT DATALATTICE statement creates a grid of graphs based on one or two classification variables and a graphical prototype. The variable is treated as discrete. By default, a separate instance of the prototype (a data cell) is created for each possible combination of the classification variables. In this example, we use treatment group as the column variable to classify plots by treatment arms. The LAYOUT PROTOTYPE is a special layout used by LAYOUT DATALATTICE to define the cell, it defines the layered composite plot to be populated in each cell. Please note that the Columnweight=Proportional is a new statement in SAS 9.4.

```
proc template;
  define statgraph wthtPanel;
    begingraph;
      entrytitle 'Weight/ Height for Age Percentile';
      layout datalattice columnvar=ARM / headerlabeldisplay=value
        columnweight=proportional
        columnaxisopts=(display=(tickvalues)
          discreteopts=(tickvaluefitpolicy=none))
        rowaxisopts=(offsetmin=0)
        row2axisopts=(offsetmin=0);

      layout prototype;
        barchart x=VISIT y=WTPCT / discreteoffset=-0.2 name='WT'
          legendlabel="Wtpct"
          barwidth=0.4 stat=mean
          fillattrs=graphdata1;

        barchart x=VISIT y=HTPCT / discreteoffset=0.2 barwidth=0.4
          legendlabel="Htpct"
          stat=mean yaxis=y2
          fillattrs=graphdata2 name='PT';

      endlayout;
      sidebar / spacefill=false;
        discretelegend 'WT' 'PT';
      endsidebar;
    endlayout;
  endgraph;
end;
run;
```

```
proc sgrender data=ANA.WTA template=wthtPanel;  
run;
```

CONCLUSION

The starting point for this paper has been how to create graphs with multiple Y axes, three types of graphs which are commonly used in our daily work has been shown. As new features have been added to the SG procedures and TEMPLATE procedures in SAS 9.4, more complex and informative graphics can be created in a flexible way. Understanding of the simple and powerful graphics features of SAS 9.4 will empower the programmers to manipulate and customize graphs that can fully support clinical data review and decision making.

REFERENCES

Complex Clinical Graphs using SAS, Sanjay Matange. SAS Institute, Inc.
SAS(R) 9.4 Graph Template Language: Reference, Fourth Edition. Available at:
<https://support.sas.com/documentation/cdl/en/grstatgraph/67882/HTML/default/viewer.htm#n1dlakx61v72in1k3ebm8rz18qd.htm>

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Mina Chen
Enterprise: Roche (China) Holding Ltd.
Address: 6th Floor, 3rd Tower, German Center, No. 88, Ke Yuan Road
City, State ZIP: Shanghai, China
Work Phone: +86-021-2892 3475, 201203
E-mail: mina.chen@roche.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.